/**
 * Copyright (c) 2009 Teoti Graphix, LLC See:
 *     http://tkarchitecture.teotigraphix.org
 * 
 * Permission is hereby granted, free of charge, to any person
 * obtaining a copy of this software and associated documentation
 * files (the "Software"), to deal in the Software without
 * restriction, including without limitation the rights to use,
 * copy, modify, merge, publish, distribute, sublicense, and/or sell
 * copies of the Software, and to permit persons to whom the
 * Software is furnished to do so, subject to the following
 * conditions:
 * 
 * The above copyright notice and this permission notice shall be
 * included in all copies or substantial portions of the Software.
 * 
 * THE SOFTWARE IS PROVIDED "AS IS", WITHOUT WARRANTY OF ANY KIND,
 * EXPRESS OR IMPLIED, INCLUDING BUT NOT LIMITED TO THE WARRANTIES
 * OF MERCHANTABILITY, FITNESS FOR A PARTICULAR PURPOSE AND
 * NONINFRINGEMENT. IN NO EVENT SHALL THE AUTHORS OR COPYRIGHT
 * HOLDERS BE LIABLE FOR ANY CLAIM, DAMAGES OR OTHER LIABILITY,
 * WHETHER IN AN ACTION OF CONTRACT, TORT OR OTHERWISE, ARISING
 * FROM, OUT OF OR IN CONNECTION WITH THE SOFTWARE OR THE USE OR
 * OTHER DEALINGS IN THE SOFTWARE.
 * 
 * @author Michael Schmalle mschmalle@teotigraphix.com
 */
package org.example.ui.behavior
{

import flash.events.IEventDispatcher;

import org.example.ui.component.Button;
import org.teotigraphix.dx.interfaces.IValidatingBehavior;
import org.teotigraphix.tka.patterns.behavior.SparkComponentBehavior;

import spark.components.supportClasses.TextBase;

/**
 * Adds label truncation behavior to a <code>Button</code>.
 * 
 * <p>Note : This is a proof of concept abstraction; this could just as esily 
 * be implemented in the <code>Button.coreBehavior</code>. The behavior also
 * shows how a non-core behavior can pipe into the Component's validation
 * through implementing the <code>IValidatingBehavior</code> interface.</p>
 * 
 * @copyright Teoti Graphix, LLC
 * @author Michael Schmalle
 * @date 07-15-09
 */
public class TruncatedToolTipBehavior 
	   extends SparkComponentBehavior
	   implements IValidatingBehavior
{
	//--------------------------------------------------------------------------
	//
	//  Public SkinPart :: Variables
	//
	//--------------------------------------------------------------------------
	
	[SkinPartAlias]
	
	/**
	 * The label part of a button.
	 */
	public var labelDisplay:TextBase;
	
	//--------------------------------------------------------------------------
	//
	//  Public ComponentAlias :: Variables
	//
	//--------------------------------------------------------------------------
	
	[ComponentAlias]
	
	/**
	 * The spark button cast.
	 */
	public var button:Button;
	
	//--------------------------------------------------------------------------
	//
	//  Constructor
	//
	//--------------------------------------------------------------------------
	
	/**
	 * Constructor.
	 */
	public function TruncatedToolTipBehavior()
	{
		super();
	}
	
	//--------------------------------------------------------------------------
	//
	//  IValidateBehavior API :: Methods
	//
	//--------------------------------------------------------------------------
	
	/**
	 * @copy IValidateBehavior#validateProperties()
	 */
	public function validateProperties():void
	{
	}
	
	/**
	 * @copy IValidateBehavior#validateSize()
	 */
	public function validateSize():void
	{
	}
	
	/**
	 * @copy IValidateBehavior#validateDisplayList()
	 */
	public function validateDisplayList(unscaledWidth:Number,
										unscaledHeight:Number):void
	{
		checkLabelDisplayTruncation();
	}
	
	/**
	 * @copy IValidateBehavior#validateStyle()
	 */
	public function validateStyle(styleProp:String):void
	{
	}
	
	//--------------------------------------------------------------------------
	//
	//  Overridden Protected :: Methods
	//
	//--------------------------------------------------------------------------
	
	/**
	 * @private
	 */
	override protected function addingBehavior(instance:IEventDispatcher):void
	{
		super.addingBehavior(instance);
	}	
	
	/**
	 * @private
	 */
	override protected function removingBehavior(instance:IEventDispatcher):void
	{
		button.hoverToolTip = null;
		
		super.removingBehavior(instance);
	}
	
	//--------------------------------------------------------------------------
	//
	//  Protected :: Methods
	//
	//--------------------------------------------------------------------------
	
	/**
	 * Checks the label displays need for truncation.
	 */
	protected function checkLabelDisplayTruncation():void
	{
		if (!labelDisplay || button.explicitToolTip)
			return;
		
		labelDisplay.validateNow();
		var isTruncated:Boolean = labelDisplay.isTruncated;
		
		button.hoverToolTip = isTruncated ? labelDisplay.text : null;
	}
}
}